

**estimates** — Save and manipulate estimation results

Syntax    Description    Remarks and examples    Also see

## Syntax

| Command   | Reference                              |
|---|--|
| <i>Save and use results from disk</i>                 |  |
| <code>estimates save <i>filename</i></code>           | [R] <a href="#">estimates save</a>     |
| <code>estimates use <i>filename</i></code>            | [R] <a href="#">estimates save</a>     |
| <code>estimates describe using <i>filename</i></code> | [R] <a href="#">estimates describe</a> |
| <code>estimates esample: ...</code>                   | [R] <a href="#">estimates save</a>     |
| <hr/>   |  |
| <i>Store and restore estimates in memory</i>          |  |
| <code>estimates store <i>name</i></code>              | [R] <a href="#">estimates store</a>    |
| <code>estimates restore <i>name</i></code>            | [R] <a href="#">estimates store</a>    |
| <code>estimates query</code>                          | [R] <a href="#">estimates store</a>    |
| <code>estimates dir</code>                            | [R] <a href="#">estimates store</a>    |
| <code>estimates drop <i>namelist</i></code>           | [R] <a href="#">estimates store</a>    |
| <code>estimates clear</code>                          | [R] <a href="#">estimates store</a>    |
| <hr/>   |  |
| <i>Set titles and notes</i>                           |  |
| <code>estimates title: <i>text</i></code>             | [R] <a href="#">estimates title</a>    |
| <code>estimates title</code>                          | [R] <a href="#">estimates title</a>    |
| <code>estimates notes: <i>text</i></code>             | [R] <a href="#">estimates notes</a>    |
| <code>estimates notes</code>                          | [R] <a href="#">estimates notes</a>    |
| <code>estimates notes list ...</code>                 | [R] <a href="#">estimates notes</a>    |
| <code>estimates notes drop ...</code>                 | [R] <a href="#">estimates notes</a>    |
| <hr/>   |  |
| <i>Report</i>   |  |
| <code>estimates describe [<i>name</i>]</code>         | [R] <a href="#">estimates describe</a> |
| <code>estimates replay [<i>namelist</i>]</code>       | [R] <a href="#">estimates replay</a>   |
| <hr/>   |  |
| <i>Tables and statistics</i>                          |  |
| <code>estimates table [<i>namelist</i>]</code>        | [R] <a href="#">estimates table</a>    |
| <code>estimates stats [<i>namelist</i>]</code>        | [R] <a href="#">estimates stats</a>    |
| <code>estimates for <i>namelist</i>: ...</code>       | [R] <a href="#">estimates for</a>      |

## Description

`estimates` allows you to store and manipulate estimation results:

- You can save estimation results in a file for use in later sessions.
- You can store estimation results in memory so that you can
  - a. switch among separate estimation results and
  - b. form tables combining separate estimation results.

## Remarks and examples

[stata.com](http://www.stata.com)

`estimates` is for use after you have fit a model, be it with `regress`, `logistic`, etc. You can use `estimates` after any estimation command, whether it be an official estimation command of Stata or a user-written one.

`estimates` has three separate but related capabilities:

1. You can save estimation results in a file on disk so that you can use them later, even in a different Stata session.
2. You can store up to 300 estimation results in memory so that they are at your fingertips.
3. You can make tables comparing any results you have stored in memory.

Remarks are presented under the following headings:

*[Saving and using estimation results](#)*  
*[Storing and restoring estimation results](#)*  
*[Comparing estimation results](#)*  
*[Jargon](#)*

## Saving and using estimation results

After you have fit a model, say, with `regress`, type

```
. use http://www.stata-press.com/data/r13/auto
(1978 Automobile Data)
. regress mpg weight displ foreign
(output omitted)
```

You can save the results in a file:

```
. estimates save basemodel
(file basemodel.ster saved)
```

Later, say, in a different session, you can reload those results:

```
. estimates use basemodel
```

The situation is now nearly identical to what it was immediately after you fit the model. You can replay estimation results:

```
. regress
(output omitted)
```

You can perform tests:

```
. test foreign==0
(output omitted)
```

And you can use any postestimation command or postestimation capability of Stata. The only difference is that Stata no longer knows what the estimation sample, `e(sample)` in Stata jargon, was. When you reload the estimation results, you might not even have the original data in memory. That is okay. Stata will know to refuse to calculate anything that can be calculated only on the original estimation sample.

If it is important that you use a postestimation command that can be used only on the original estimation sample, there is a way you can do that. You use the original data and then use `estimates esample:` to tell Stata what the original sample was.

See [R] [estimates save](#) for details.

## Storing and restoring estimation results

Storing and restoring estimation results in memory is much like saving them to disk. You type

```
. estimates store base
```

to save the current estimation results under the name `base`, and you type

```
. estimates restore base
```

to get them back later. You can find out what you have stored by typing

```
. estimates dir
```

Saving estimation results to disk is more permanent than storing them in memory, so why would you want merely to store them? The answer is that, once they are stored, you can use other `estimates` commands to produce tables and reports from them.

See [R] [estimates store](#) for details about the `estimates store` and `restore` commands.

## Comparing estimation results

Let's say that you have done the following:

```
. use http://www.stata-press.com/data/r13/auto
(1978 Automobile Data)
. regress mpg weight displ
(output omitted)
. estimates store base
. regress mpg weight displ foreign
(output omitted)
. estimates store alt
```

You can now get a table comparing the coefficients:

```
. estimates table base alt
```

| Variable     | base       | alt        |
|--------------|------------|------------|
| weight       | -.00656711 | -.00677449 |
| displacement | .00528078  | .00192865  |
| foreign      |            | -1.6006312 |
| _cons        | 40.084522  | 41.847949  |

`estimates table` can do much more; see [R] [estimates table](#). Also see [R] [estimates stats](#). `estimates stats` works similarly to `estimates table` but produces model comparisons in terms of BIC and AIC.

### Jargon

You know that if you fit a model, say, by typing

```
. regress mpg weight displacement
```

then you can later replay the results by typing

```
. regress
```

and you can do tests and calculate other postestimation statistics by typing

```
. test displacement==0
. estat vif
. predict mpghat
```

As a result, we often refer to the *estimation results* or the *current estimation results* or the *most recent estimation results* or the *last estimation results* or the *estimation results in memory*.

With `estimates store` and `estimates restore`, you can have many estimation results in memory. One set of those, the set most recently estimated, or the set most recently restored, are the *current* or *active* estimation results, which you can replay, which you can test, or from which you can calculate postestimation statistics.

*Current* and *active* are the two words we will use interchangeably from now on.

### Also see

[P] [\\_estimates](#) — Manage estimation results